QWT_ROOT = C:/Qwt-6.0.2
PRO_ROOT = D:/projects/qt_proj

OBJECTS_DIR = $${PRO_ROOT}/temp/$${set_app_name}
MOC_DIR = $${OBJECTS_DIR}

#application type
contains(set_app_type, staticlib) {
    message(Building staticlib ...)
    TEMPLATE = lib
    CONFIG += staticlib
} else {
    contains(set_app_type, binary){
        message(Building binary ...)
        TEMPLATE = app
    }
}

CONFIG(debug, debug|release) {
    DESTDIR = $${PRO_ROOT}/$${TEMPLATE}/debug
    set_mode = debug
    qwt_lib = qwtd
} else {
    DESTDIR = $${PRO_ROOT}/$${TEMPLATE}/release
    set_mode = release
    qwt_lib = qwt

}


###############Section: include headers and libs##################
INCLUDEPATH += include

for(elem, set_inc_libs) {
    message(INCLUDE libs $$elem)
    INCLUDEPATH += $${PRO_ROOT}/$${elem}/include
    LIBS += -L$${PRO_ROOT}/lib/$${set_mode} -l$${elem}
}
###############Section: End######################################

###############Section: qwt setting #############################

contains(set_3rd_libs, qwt) {
    message(including qwt ...)

    CONFIG += qwt
    INCLUDEPATH += $${QWT_ROOT}/include
    LIBS += -L$${QWT_ROOT}/lib -l$${qwt_lib}
}

###############Section: End######################################



message(incs $${INCLUDEPATH})
message(libs $${LIBS})
